

Example 3b: Fixed and Random Effects in General Multilevel Models for Two-Level Nested Outcomes (as estimated using restricted maximum likelihood in SAS MIXED and STATA MIXED)

This example uses real data from a math test given at the end of 10th grade in a Midwestern Rectangular State. These analyses include 13,802 students from 94 schools, with 31–515 students in each school ($M = 275$). We will examine how student free and reduced lunch status (0 = pay for lunch, 1 = receive free or reduced lunch) predicts student math test scores.

SAS Syntax for Data Import, Manipulation, and Description:

```
* Define global variable for file location to be replaced in code below;
* \\Client\ precedes actual path when using UIowa Virtual Desktop;
%LET filesave=C:\Dropbox\19_PSQF7375_Clustered\PSQF7375_Clustered_Example3b;
LIBNAME example "&filesave.";

* Import data into work library;
DATA work.grade10; SET example.grade10school;
  LABEL studentID= "studentID: Student ID number"
        schoolID= "schoolID: School ID number"
        frlunch= "frlunch: 0=No, 1=Free/Reduced Lunch"
        math= "math: Math Test Score Outcome";
  * Selecting cases that are complete for analysis variables;
  IF NMISS(studentID, schoolID, frlunch, math)>0 THEN DELETE; RUN;

* Get school means;
PROC SORT DATA=work.grade10; BY schoolID studentID; RUN;
PROC MEANS NOPRINT N DATA=work.grade10;
  BY schoolID; VAR frlunch math;
  OUTPUT OUT=work.SchoolMeans MEAN(frlunch math)= SMfrlunch SMmath; RUN;

* Label new school mean variables;
DATA work.SchoolMeans; SET work.SchoolMeans;
  Nperschool = _FREQ_; * Saving N per school;
  DROP _TYPE_ _FREQ_; * Dropping unneeded SAS-created variables;
  LABEL Nperschool= "Nperschool: # Students Contributing Data"
        SMfrlunch= "SMfrlunch: School Mean 0=No, 1=Free/Reduced Lunch"
        SMmath= "SMmath: School Mean Math Outcome";
  * Arbitrarily select only schools with at least 30 students;
  IF Nperschool < 31 THEN DELETE;
  * Center school mean predictor;
  SMfrlunch30 = SMfrlunch - .30;
  LABEL SMfrlunch30= "SMfrlunch30: 0=.30)"; RUN;

* Merge school means back with individual data;
DATA work.grade10; MERGE work.grade10 work.SchoolMeans; BY schoolID;
  * Arbitrarily select only schools with at least 30 students;
  IF Nperschool < 31 THEN DELETE; RUN;

* Sort in order of ID variables;
PROC SORT DATA=work.grade10;
  BY schoolID studentID;
RUN;

TITLE "School-Level Descriptives";
PROC MEANS NDEC=2 DATA=work.SchoolMeans;
  VAR Nperschool SMmath SMfrlunch;
RUN; TITLE;

TITLE "Student-Level Descriptives";
PROC MEANS NDEC=2 DATA=work.grade10;
  VAR math frlunch;
RUN; TITLE;
```

STATA Syntax and Output for Data Import, Manipulation, and Description:

```

/ Define global variable for file location to be replaced in code below
// \\Client\ precedes actual path when using UIowa Virtual Desktop
global filesave "C:\Dropbox\19_PSQF7375_Clustered\PSQF7375_Clustered_Example3b"

// Import example stata data file
use "$filesave\grade10school.dta", clear

// Label existing variables
label variable studentID "studentID: Student ID number"
label variable schoolID "schoolID: School ID number"
label variable frlunch "frlunch: Student Free/Reduced Lunch 0=No 1=Yes"
label variable math "math: Student Free/Reduced Lunch 0=No 1=Yes"

// Get school means of variables and label them
egen SMfrlunch = mean(frunch), by (schoolID)
egen SMmath = mean(math), by (schoolID)
label variable SMfrlunch "SMfrlunch: School Mean 0=No, 1=Free/Reduced Lunch"
label variable SMmath "SMmath: School Mean Math Outcome"

// Get number of students per school
egen Nperschool = count(studentID), by (schoolID)
label variable Nperschool "Nperschool: # Students Contributing Data"

// Center school mean predictor
gen SMfrlunch30 = SMfrlunch - .30
label variable SMfrlunch30 "SMfrlunch30: % Students with Free Lunch (0=30%)"

// Drop schools with <= 30 students
drop if Nperschool < 31

display as result "STATA School-Level Descriptives"
preserve // Save for later use, then compute school-level dataset
collapse Nperschool SMfrlunch SMmath, by(schoolID)
format Nperschool SMfrlunch SMmath %4.2f
summarize Nperschool SMfrlunch SMmath, format

```

Variable	Obs	Mean	Std. Dev.	Min	Max
Nperschool	94	139.17	138.20	31.00	515.00
SMfrlunch	94	0.30	0.21	0.00	0.80
SMmath	94	47.73	6.97	29.45	61.61

```

restore // Go back to student-level dataset
display as result "STATA Student-Level Descriptives"
format math frlunch %4.2f
summarize math frlunch, format

```

Variable	Obs	Mean	Std. Dev.	Min	Max
math	13,082	48.12	17.26	0.00	83.00
frlunch	13,082	0.31	0.46	0.00	1.00

Model 1: Two-Level Empty Means, Random Intercept for Math Outcome (for pupil p in school s)

$$\begin{array}{l} \text{Level 1: } \text{Math}_{ps} = \beta_{0s} + e_{ps} \\ \text{Level 2: } \beta_{0s} = \gamma_{00} + U_{0s} \end{array}$$

```

TITLE "SAS Model 1: 2-Level Empty Means, Random Intercept for Math Outcome";
PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS schoolID; * Put nesting variable on CLASS to speed estimation;
  MODEL math = / SOLUTION DDFM=Satterthwaite OUTPM=PredEmpty;
  * Asking for G for all, V and VCORR for first school in order of schoolID;
  RANDOM INTERCEPT / G V=1 VCORR=1 TYPE=UN SUBJECT=schoolID; * VCORR gives ICC;
  * Asking for R for first school in order of schoolID;
  REPEATED / R=1 TYPE=VC SUBJECT=schoolID; * Default R matrix is diagonal (VC);
  * ODS saves results for pseudo-R2 macro;
  ODS OUTPUT CovParms=CovEmpty; RUN; TITLE;

display as result "STATA Model 1: 2-Level Empty Means, Random Intercept for Math Outcome"
mixed math , ///
  || schoolID: , variance reml covariance(un) dfmethod(satterthwaite) ///
  dftable(pvalue) residuals(independent), // residuals: diagonal R matrix default
  estat ic, n(94), // Get Information Criteria
  estat icc // Get Intraclass Correlation
  estat recovariance, releval(schoolID) // Get G matrix for whole sample
  estat wcorrelation, covariance at(schoolID=125) // Get V matrix for first schoolID
  estat wcorrelation, at(schoolID=125) // Get VCORR matrix for first schoolID

```

SAS output:

Truncated R Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
1	253.16			
2		253.16		
3			253.16	
4				253.16

This diagonal (TYPE=VC for “variance components”) **level-1 R** matrix depicts the assumption of no residual covariance across students from the same school after controlling for the school random intercept (whose variance is shown in the **level-2 G** matrix, next).

Estimated G Matrix (for whole sample)

Row	Effect	schoolID	Col1
1	Intercept	125	45.3948

The **level-2 G** matrix holds the variances and covariances of the school-level random effects, which will ALWAYS be TYPE=UN (unstructured). Here, **G** is a 1x1 matrix, or scalar.

Truncated V Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
1	298.55	45.3682	45.3682	45.3682
2	45.3682	298.55	45.3682	45.3682
3	45.3682	45.3682	298.55	45.3682
4	45.3682	45.3682	45.3682	298.55

The combined predicted variance and covariance across persons for one school is shown here in **V**, which has a specific form—“compound symmetry”—that results from having only a random intercept variance and a diagonal **R** matrix with the same level-1 residual variance for all persons on the diagonal.

Truncated V Correlation Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
1	1.0000	0.1520	0.1520	0.1520
2	0.1520	1.0000	0.1520	0.1520
3	0.1520	0.1520	1.0000	0.1520
4	0.1520	0.1520	0.1520	1.0000

VCORR is the correlation version of V, which provides the ICC on the off-diagonal (constant correlation across persons).

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	schoolID	45.3682	7.1288	6.36	<.0001
Residual		253.18	3.1416	80.59	<.0001

Calculate the ICC for the correlation of students in the same school for math:

$$ICC = \frac{45.37}{45.37 + 253.18} = .152$$

Null Model Likelihood Ratio Test			This LR test tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.			
DF	Chi-Square	Pr > ChiSq				
1	1860.21	<.0001				
Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
109790	2	109794	109794	109796	109799	109801
Solution for Fixed Effects						
	Estimate	Standard Error	DF	t Value	Pr > t	
Effect						
Intercept	47.7559	0.7224	94.2	66.11	<.0001	

Design effect using mean #students per school: $= 1 + ((n - 1) * ICC) \rightarrow 1 + [(275-1)*.152] = 41.80$

Effective sample size: $N_{\text{effective}} = (\text{\#Total Obs}) / \text{Design Effect} \rightarrow 13,082 / 42.1 = 312!!!$

95% random effect confidence interval for the intercept across schools:

Fixed effect $\pm 1.96 * \text{SQRT}(\text{random variance})$

$47.76 \pm 1.96 * \text{SQRT}(45.37) = 34.55 \text{ to } 60.96$

\rightarrow 95% of our sample's schools are predicted to have school mean math from 34.55 to 60.96

Model 2: Adding a Fixed Effect of Student Free/Reduced Lunch (Level 1)

$$\begin{aligned} \text{Level 1: } \text{Math}_{ps} &= \beta_{0s} + \beta_{1s} (\text{FRLunch}_{ps}) + e_{ps} \\ \text{Level 2: } \text{Intercept: } \beta_{0s} &= \gamma_{00} + U_{0s} \\ \text{Free/Reduced Lunch: } \beta_{1s} &= \gamma_{10} \end{aligned}$$

```
TITLE "SAS Model 2: Add Fixed Effect of Student Free/Reduced Lunch";
PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS schoolID;
  MODEL math = frlunch / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
  * ODS saves results for pseudo-R2 and LRT macros;
  ODS OUTPUT CovParms=CovFR1 InfoCrit=FitFR1; RUN; TITLE;
```

```
display as result "STATA Model 2: Add Fixed Effect of Student Free/Reduced Lunch"
mixed math c.frlunch, ///
  || schoolID: , variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue),
  estat ic, n(94)
```

SAS output:

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	schoolID	27.2239	4.5119	6.03	<.0001	
Residual		239.35	2.9703	80.58	<.0001	
Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
109016	2	109020	109020	109022	109025	109027
Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept	50.6148	0.5797	96.9	87.31	<.0001	
frlunch	-9.4288	0.3318	13E3	-28.42	<.0001	

```
* Calculate PseudoR2 relative to empty model;
%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFR1);
```

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	schoolID	45.3682	7.1288	6.36	<.0001	.
CovEmpty	Residual		253.18	3.1416	80.59	<.0001	.
CovFR1	UN(1,1)	schoolID	27.2239	4.5119	6.03	<.0001	0.39993
CovFR1	Residual		239.35	2.9703	80.58	<.0001	0.05463

What does the effect of student free/reduced lunch represent in model 2?

Children who get free/reduced lunch are predicted to score 9.43 points lower in math than children who don't.

We know this level-1 fixed effect must be smushed because it accounted for level-2 random intercept variance.

What are we assuming about the effect of student free/reduced lunch in model 2?

We are assuming no contextual effect (that the between-school and within-school effects of FRlunch are equal).

Model 3: Adding a Fixed Effect of School Proportion Free/Reduced Lunch (Level 2)

$$\begin{aligned} \text{Level 1: } \text{Math}_{ps} &= \beta_{0s} + \beta_{1s} (\text{FRlunch}_{ps}) + e_{ps} \\ \text{Level 2: } \text{Intercept: } \beta_{0s} &= \gamma_{00} + \gamma_{01} (\overline{\text{SchoolFRLunch}_s} - .30) + U_{0s} \\ \text{Free/Reduced Lunch: } \beta_{1s} &= \gamma_{10} \end{aligned}$$

```
TITLE "SAS Model 3: Add Fixed Effect of School Proportion Free/Reduced Lunch";
PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS schoolID;
  MODEL math = frlunch SMfrlunch30 / SOLUTION DDFM=Satterthwaite OUTPM=work.PredLunch;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
  ODS OUTPUT CovParms=CovFR2 InfoCrit=FitFR2;
  CONTRAST "Test of Model R2 (Omnibus FR Lunch)" frlunch 1, SMfrlunch30 1;
  ESTIMATE "FR Lunch Between-School Effect" frlunch 1 SMfrlunch30 1; RUN; TITLE;

display as result "STATA Model 3: Add Fixed Effect of School Proportion Free/Reduced Lunch"
mixed math c.frlunch c.SMfrlunch30, ///
  || schoolID: , variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue),
  estat ic, n(94),
  lincom 1*c.frlunch + 1*c.SMfrlunch30, small // FR lunch between-school effect
  estimates store FixFRLunch, // save LL for LRT
  predict predlunch, // save fixed-effect predicted outcomes
  corr math predlunch
  display as result r(rho)^2 // total R2
```

SAS output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	schoolID	13.8884	2.6315	5.28	<.0001
Residual		239.41	2.9718	80.56	<.0001

Information Criteria					
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC CAIC
108962	2	108966	108966	108968	108971 108973

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	50.6034	0.4392	91.7	115.22	<.0001
frlunch	-9.1729	0.3344	13E3	-27.43	<.0001
SMfrlunch30	-16.8434	2.0251	83.3	-8.32	<.0001

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
FR Lunch Between-School Effect	-26.0162	1.9973	78.8	-13.03	<.0001

What does the effect of school proportion free/reduced lunch represent in model 3?

This is the level-2 contextual effect for FRlunch: holding child lunch status constant, for every 10% more children in your school who get free/reduced lunch, school mean math is predicted to be lower by 1.68 points. Without controlling for individual kid lunch status, the reduction is 2.60 points per 10% (the level-2 between-school effect, given in estimate requested separately).

What does the effect of student free/reduced lunch NOW represent in model 3?

This is the pure within-school effect: holding school lunch status constant, children who receive free/reduced lunch are predicted to score 9.17 points lower in math than children who don't.

* Calculate PseudoR2 relative to previous model 2;
`%PseudoR2(NCov=2, CovFewer=CovFR1, CovMore=CovFR2);`

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovFR1	UN(1,1)	schoolID	27.2239	4.5119	6.03	<.0001	.
CovFR1	Residual		239.35	2.9703	80.58	<.0001	.
CovFR2	UN(1,1)	schoolID	13.8884	2.6315	5.28	<.0001	0.48984
CovFR2	Residual		239.41	2.9718	80.56	<.0001	-0.00027

Total reduction from **both** lunch effects:
 Intercept variance → 69.39% (of 15.2%)
 Residual variance → 5.44% (of 84.8%)

* Calculate PseudoR2 relative to empty model 1 (total for FRlunch);
`%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFR2);`

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	schoolID	45.3682	7.1288	6.36	<.0001	.
CovEmpty	Residual		253.18	3.1416	80.59	<.0001	.
CovFR2	UN(1,1)	schoolID	13.8884	2.6315	5.28	<.0001	0.69387
CovFR2	Residual		239.41	2.9718	80.56	<.0001	0.05438

* Calculate TotalR2 relative to empty model 1 (total for FRlunch);
`%TotalR2(DV=math, PredFewer=PredEmpty, PredMore=PredLunch);`

Name	Pred Corr	TotalR2	Total R2Diff
PredEmpty	0.00000	0.00000	.
PredLunch	0.40382	0.16307	0.16307

This total- $R^2 = .16$ is significantly > 0 according to the multivariate Wald test below for the two FR lunch fixed effects, $F(2,164) = 461.00, p < .0001$.

Contrasts				
Label	Num DF	Den DF	F Value	Pr > F
Test of Model R2 (Omnibus FR Lunch)	2	164	461.01	<.0001

Model 4: Adding a Random Effect of Student Free/Reduced Lunch (over Schools)

$$\begin{aligned} \text{Level 1: } \text{Math}_{ps} &= \beta_{0s} + \beta_{1s} (\text{FRLunch}_{ps}) + e_{ps} \\ \text{Level 2: } \text{Intercept: } \beta_{0s} &= \gamma_{00} + \gamma_{01} (\text{SchoolFRLunch}_s - .30) + U_{0s} \\ \text{Free/Reduced Lunch: } \beta_{1s} &= \gamma_{10} + U_{1s} \end{aligned}$$

```

TITLE "SAS Model 4: Add Random Effect of Student Free/Reduced Lunch";
PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
CLASS schoolID;
MODEL math = frlunch SMfrlunch30 / SOLUTION DDFM=Satterthwaite;
* Asking for G and GCORR for all, V and VCORR for first school in order of schoolID;
RANDOM INTERCEPT frlunch / G GCORR V=1 VCORR=1 TYPE=UN SUBJECT=schoolID;
* Asking for R for first school in order of schoolID;
REPEATED / R=1 TYPE=VC SUBJECT=schoolID; * Default R matrix is diagonal (VC);
ODS OUTPUT CovParms=CovFR2RandFR1 InfoCrit=FitFR2RandFR1; RUN; TITLE;

display as result "STATA Model 4: Add Random Effect of Student Free/Reduced Lunch"
mixed math c.frlunch c. SMfrlunch30, ///
|| schoolID: c.frlunch, variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue),
estat ic, n(94),
estat recovariance, relevel(schoolID) // Get G matrix for whole sample
estat recovariance, relevel(schoolID) correlation // Get GCORR matrix for whole sample
estat wcorrelation, covariance at(schoolID=125) // Get V matrix for first schoolID
estat wcorrelation, at(schoolID=125) // Get VCORR matrix for first schoolID
estimates store RandFRLunch // save LL for LRT
lrtest RandFRLunch FixFRLunch // LRT against fixed effect model

```

SAS output:

Truncated R Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
1	236.83			
2		236.83		
3			236.83	
4				256.83

This diagonal (TYPE=VC for “variance components”) **level-1 R** matrix depicts the same assumption of no residual covariance across students from the same school after controlling for the school random intercept AND the school random FRLunch slope (whose variances and covariance are shown in the level-2 **G** matrix, next).

Estimated G Matrix (for whole sample)

Row	Effect	schoolID	Col1	Col2
1	Intercept	125	20.3758	-12.0351
2	frlunch	125	-12.0351	12.9493

The level-2 **G** matrix holds the variances and covariances of the school-level random effects, which will ALWAYS be TYPE=UN (unstructured). Now, **G** is a 2x2 matrix adding the random variance of the level-1 FRLunch effect (and its covariance with the random intercept). GCORR provides the correlation version.

Estimated G Correlation Matrix (for whole sample)

Row	Effect	number	Col1	Col2
1	Intercept	125	1.0000	-0.7409
2	frlunch	125	-0.7409	1.0000

Truncated V Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
FR=1 1	246.09	8.3407	8.3407	9.2550
FR=0 2	8.3407	257.21	20.3758	8.3407
FR=0 3	8.3407	20.3758	257.21	8.3407
FR=1 4	9.2550	8.3407	8.3407	246.09

The combined predicted variance and covariance across persons for one school is shown here in **V**, which now shows that the predicted variance and covariance across students differs as a function of student FRLunch. Consequently the ICC differs—it’s now conditional.

Truncated V Correlation Matrix for schoolID 125 (is actually 35x35)

Row	Col1	Col2	Col3	Col4
FR=1 1	1.0000	0.03315	0.03315	0.03761
FR=0 2	0.03315	1.0000	0.07922	0.03315
FR=0 3	0.03315	0.07922	1.0000	0.03315
FR=1 4	0.03761	0.03315	0.03315	1.0000

VCORR is the correlation version of V, which provides the conditional ICC on the off-diagonal (non-constant correlation across persons).

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	schoolID	20.3739	3.8391	5.31	<.0001 L2 random intercept variance
UN(2,1)	schoolID	-12.0368	3.2255	-3.73	0.0002 L2 intercept-lunch covariance
UN(2,2)	schoolID	12.9489	3.3833	3.83	<.0001 L2 random slope variance for frlunch
Residual		236.84	2.9467	80.37	<.0001 L1 residual variance

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
108874	4	108882	108882	108886	108892	108896

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	50.2568	0.5192	83.7	96.79	<.0001
frlunch	-8.4458	0.5647	97.3	-14.96	<.0001
SMfrlunch30	-17.0938	1.9434	76.6	-8.80	<.0001

* Calculate difference in model fit relative to fixed-FRlunch-only model 3;

```
%FitTest(FitFewer=FitFR2, FitMore=FitFR2RandFR1);
```

Likelihood Ratio Test for FitFR2 vs. FitFR2RandFR1

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFR2	108962	2	108966	108971	.	.	.
FitFR2RandFR1	108874	4	108882	108892	88.2545	2	0

Is model 4 better than model 3?

Yes, $-2\Delta LL(2) = 88, p < .0001$

So what does this mean about the effect of student free/reduced lunch? *The difference in math between kids who get free/reduced lunch and kids who don't varies significantly over schools.*

95% random effects CI for the random FRlunch slope: $\rightarrow -8.45 \pm 1.96 * \text{SQRT}(12.95) = -15.50 \text{ to } -1.39$
On average, the gap in math related to lunch status is 8.45 points, but across 95% of the schools, that gap is predicted to be anywhere from 1.39 to 15.50 points. Now we need to explain it!

Model 5: Adding a Cross-Level Interaction of Student by School Free/Reduced Lunch

$$\begin{aligned} \text{Level 1: } \text{Math}_{ps} &= \beta_{0s} + \beta_{1s}(\text{FRlunch}_{ps}) + e_{ps} \\ \text{Level 2: } \text{Intercept: } \beta_{0s} &= \gamma_{00} + \gamma_{01}(\overline{\text{SchoolFRLunch}_s} - .30) + U_{0s} \\ \text{Free/Reduced Lunch: } \beta_{1s} &= \gamma_{10} + \gamma_{11}(\overline{\text{SchoolFRLunch}_j} - .30) + U_{1s} \end{aligned}$$

TITLE "SAS Model 5: Add Cross-Level Interaction of Student by School Free/Reduced Lunch";

PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;

CLASS schoolID;

MODEL math = frlunch SMfrlunch30 frlunch*SMfrlunch30

/ SOLUTION DDFM=Satterthwaite;

RANDOM INTERCEPT frlunch / G TYPE=UN SUBJECT=schoolID;

ODS OUTPUT CovParms=CovInt1 InfoCrit=FitInt1; RUN; TITLE;

display as result "STATA Model 5: Add Cross-Level Interaction of Student by School FR Lunch"

mixed math c.frlunch c.SMfrlunch30 c.frlunch#c.SMfrlunch30, ///

|| schoolID: c.frlunch, variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue), estat ic, n(94)

SAS output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	schoolID	20.3631	3.8212	5.33	<.0001
UN(2,1)	schoolID	-11.6039	3.1876	-3.64	0.0003
UN(2,2)	schoolID	12.2630	3.2958	3.72	<.0001
Residual		236.82	2.9464	80.38	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
108867	4	108875	108875	108880	108886	108890

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	50.2191	0.5197	84.9	96.63	<.0001
frlunch	-8.6779	0.5737	110	-15.13	<.0001
SMfrlunch30	-19.4657	2.5002	88.1	-7.79	<.0001
frlunch*SMfrlunch30	4.1449	2.6655	101	1.56	0.1231

What does the effect of student free/reduced lunch NOW represent in model 5?

This is the difference between kids who get free/reduced lunch and those who don't in schools where 30% of the kids get free/reduced lunch: those kids who get free/reduced lunch are predicted to be lower in math by 8.68.

What does the effect of school proportion free/reduced lunch NOW represent in model 5? *This is the level-2 contextual (incremental between-school) effect for a kid who does not receive FR lunch: for every 10% more kids in their school that receive FR lunch, their school mean math is predicted to be lower by 1.95.*

What does the cross-level interaction of student by school free/reduced lunch represent in model 5?

The effect of being a kid who receives free/reduced lunch is reduced nonsignificantly by 0.41 for every 10% more children in their school who get free/reduced lunch. But this effect is currently smushed—it assumes without testing that school FRlunch moderates the within-school and between-school effects of FRlunch to the same extent. We know this because the interaction also reduced the level-2 random intercept variance.

*** Calculate PseudoR2 for interaction relative to random FRlunch model 4;**
%PseudoR2(NCov=4, CovFewer=CovFR2RandFR1, CovMore=CovInt1);

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovFR2RandFR1	UN(1,1)	schoolID	20.3739	3.8391	5.31	<.0001	.
CovFR2RandFR1	UN(2,2)	schoolID	12.9489	3.3833	3.83	<.0001	.
CovFR2RandFR1	Residual		236.84	2.9467	80.37	<.0001	.
CovInt1	UN(1,1)	schoolID	20.3631	3.8212	5.33	<.0001	0.000531
CovInt1	UN(2,2)	schoolID	12.2630	3.2958	3.72	<.0001	0.052969
CovInt1	Residual		236.82	2.9464	80.38	<.0001	0.000048

Model 6: Adding a Level-2 Interaction of Quadratic School Free/Reduced Lunch

$$\text{Level 1: Math}_{ps} = \beta_{0s} + \beta_{1s} (\text{FRlunch}_{ps}) + e_{ps}$$

$$\text{Level 2: Intercept: } \beta_{0s} = \gamma_{00} + \gamma_{01} (\overline{\text{SchoolFRLunch}_s} - .30) + \gamma_{02} (\overline{\text{SchoolFRLunch}_j} - .30)^2 + U_{0s}$$

$$\text{Free/Reduced Lunch: } \beta_{1s} = \gamma_{10} + \gamma_{11} (\overline{\text{SchoolFRLunch}_j} - .30) + U_{1s}$$

```

TITLE "SAS Model 6: Add Level-2 Interaction of Quadratic School Free/Reduced Lunch";
PROC MIXED DATA=work.grade10 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
CLASS schoolID;
MODEL math = frlunch SMfrlunch30 frlunch*SMfrlunch30 SMfrlunch30*SMfrlunch30
          / SOLUTION DDFM=Satterthwaite OUTPM=work.PredTotal;
RANDOM INTERCEPT frlunch / G TYPE=UN SUBJECT=schoolID;
ODS OUTPUT CovParms=CovInt2 InfoCrit=FitInt2;
CONTRAST "Test of Omnibus FR Lunch Interaction" frlunch*SMfrlunch30 1, SMfrlunch30*SMfrlunch30 1;
ESTIMATE "FR Lunch Between-School Simple Main Effect" frlunch 1 SMfrlunch30 1;
ESTIMATE "FR Lunch Between-School Interaction" frlunch*SMfrlunch30 1 SMfrlunch30*SMfrlunch30 1;
RUN; TITLE;

display as result "STATA Model 6: Add Level-2 Interaction of Quadratic School Free/Reduced Lunch"
mixed math c.frlunch c.SMfrlunch30 c.frlunch#c.SMfrlunch30 c.SMfrlunch30#c.SMfrlunch30, ///
|| schoolID: c.frlunch, variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue),
estat ic, n(94),
test (c.frlunch#c.SMfrlunch30=0) (c.SMfrlunch30#c.SMfrlunch30), small // Test Omnibus Interact
lincom 1*c.frlunch + 1*c.SMfrlunch30, small // FRL BS simple main effect
lincom 1*c.frlunch#c.SMfrlunch30 + 1*c.SMfrlunch30#c.SMfrlunch30, small // FRL BS interaction
margins, at(c.frlunch=(0 1) c.SMfrlunch30=(-.2 0 .2 .4)) vsquish // create predicted values
marginsplot, noci name(predicted_lunch, replace) xdimension(frlunch) // plot predicted, no CI
predict predtotal, // save fixed-effect predicted outcomes
corr math predtotal
display as result r(rho)^2 // total R2

```

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	schoolID	19.6520	3.7262	5.27	<.0001
UN(2,1)	schoolID	-11.1877	3.1440	-3.56	0.0004
UN(2,2)	schoolID	12.3054	3.3159	3.71	0.0001
Residual		236.82	2.9463	80.38	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
108859	4	108867	108867	108871	108877	108881

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	50.8596	0.6504	101	78.20	<.0001
frlunch	-8.8242	0.5834	111	-15.13	<.0001
SMfrlunch30	-17.9760	2.6360	86.1	-6.82	<.0001
frlunch*SMfrlunch30	5.4258	2.7969	105	1.94	0.0551
SMfrlunch30*SMfrlunch30	-14.3035	8.9879	86.6	-1.59	0.1152

What does the cross-level interaction of student by school free/reduced lunch NOW represent?

The effect of being a kid who receives free/reduced lunch (now after allowing for differential moderation across levels of the effects of free/reduced lunch at both levels by school mean free/reduced lunch) is reduced significantly-ish by 0.54 for every 10% more children in their school who get free/reduced lunch.

What does the level-2 interaction of quadratic school free/reduced lunch represent?

*After controlling for kid free/reduced lunch status, the contextual (incremental between-school) effect of school mean free/reduced lunch (as evaluated at 30% FRLunch here) becomes nonsignificantly more negative by 2*1.43 for every 10% more kids in their school with free/reduced lunch.*

Label	Estimates		Standard			Pr > t
	Estimate	Error	DF	t Value		
FR Lunch Between-School Simple Main Effect	-26.8002	2.6445	88.6	-10.13		<.0001
FR Lunch Between-School Interaction	-8.8778	8.5951	74.9	-1.03		0.3050

BS Simple Effect and Interaction: *If we don't control for kid free/reduced lunch, the between-school effect of -2.68 per 10% of school mean free/reduced lunch (as evaluated at 30% FRLunch here) becomes nonsignificantly more negative by 2×0.89 for every 10% more kids in their school with free/reduced lunch.*

Conclusion: *School mean free/reduced lunch moderates the within-school FRLunch effect (by making it smaller, or less negative), but not the contextual (incremental between-school) or between-school effects.*

* Calculate PseudoR2 relative to level-1 lunch interaction only model 5;
 %PseudoR2(NCov=4, CovFewer=CovInt1, CovMore=CovInt2);

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovInt1	UN(1,1)	schoolID	20.3631	3.8212	5.33	<.0001	.
CovInt1	UN(2,2)	schoolID	12.2630	3.2958	3.72	<.0001	.
CovInt1	Residual		236.82	2.9464	80.38	<.0001	.
CovInt2	UN(1,1)	schoolID	19.6520	3.7262	5.27	<.0001	0.034920
CovInt2	UN(2,2)	schoolID	12.3054	3.3159	3.71	0.0001	-0.003453
CovInt2	Residual		236.82	2.9463	80.38	<.0001	0.000029

* Calculate PseudoR2 relative to random FRLunch model 4
 (total for FRLunch interactions);
 %PseudoR2(NCov=4, CovFewer=CovFR2RandFR1, CovMore=CovInt2);

Total reduction from **both** interactions:
 Intercept variance → 3.54%
 Lunch slope variance → 4.97%
 Residual variance → 0.00%

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovFR2RandFR1	UN(1,1)	schoolID	20.3739	3.8391	5.31	<.0001	.
CovFR2RandFR1	UN(2,2)	schoolID	12.9489	3.3833	3.83	<.0001	.
CovFR2RandFR1	Residual		236.84	2.9467	80.37	<.0001	.
CovInt2	UN(1,1)	schoolID	19.6520	3.7262	5.27	<.0001	0.035432
CovInt2	UN(2,2)	schoolID	12.3054	3.3159	3.71	0.0001	0.049698
CovInt2	Residual		236.82	2.9463	80.38	<.0001	0.000076

* Calculate TotalR2 relative to random FRLunch model 4 (total for FRLunch interactions);
 %TotalR2(DV=math, PredFewer=PredLunch, PredMore=PredTotal);

Name	Pred Corr	TotalR2	Total R2Diff
PredLunch	0.40382	0.16307	.
PredTotal	0.40512	0.16412	.001052263

This change in total- $R^2 = .16$ is not significantly > 0 according to the multivariate Wald test for the two FR interactions, $F(2,95) = 2.45$, $p = .091$.

Label	Contrasts		F Value	Pr > F
	Num DF	Den DF		
Test of Omnibus FR Lunch Interaction	2	95.2	2.45	0.0914

Sample Results Section (without “smushed” models) [indicates notes about what to customize or also include]

The extent to which student free/reduced lunch status could predict student math outcomes was examined in a series of multilevel models in which the 13,802 students were modeled as nested within their 94 schools. Residual maximum likelihood (REML) within SAS [or STATA] MIXED was used in estimating and reporting all model parameters. The significance of fixed effects was evaluated with Wald tests using Satterthwaite denominator degrees of freedom, whereas random effects were evaluated via likelihood ratio tests (i.e., $-2\Delta LL$ with degrees of freedom equal to the number of new random effects variances and covariances). Alpha was chosen as .05. Model-implied fixed effects were requested via ESTIMATE [or LINCOM] statements. Effect size was evaluated via pseudo- R^2 values for the proportion reduction in each variance component, as well as with total R^2 , the squared correlation between the actual math outcomes and the math outcomes predicted by the fixed effects.

As derived from an empty means, random intercept model, student math scores had an intraclass correlation of .152, indicating that 15.2% of the variance in math scores was between schools, a significant amount, $-2\Delta LL(1) = 1860.21$, $p < .001$. A 95% random effects confidence interval, calculated as fixed intercept $\pm 1.96 * \text{SQRT}(\text{random intercept variance})$, revealed that 95% of the sample schools were predicted to have intercepts for school mean math scores between 34.6 and 61.0. Children who did not receive free/reduced lunch were treated as the reference group. Given the large variability across schools in the proportion of students who received free/reduced lunch (from 0–80% of students), a contextual effect at level 2 was represented by the school proportion of students who receive free/reduced lunch, which was centered near the sample mean, at 30%.

The effects of free/reduced lunch status at each level were then added to the model, and together significantly reduced the total variance in math scores, total- $R^2 = .163$, $F(2,164) = 461.00$, $p < .001$. The within-school level-1 effect was significant and accounted for 5.44% of the level-1 residual variance, and indicated that students who receive free/reduced lunch are expected to have lower math scores than other students in their school by 9.17. The between-school level-2 effect was also significant and accounted for 69.4% of the level-2 random intercept variance, and indicated that for every additional 10% of students who receive free/reduced lunch, that school's mean math score is expected to be lower by 2.60. After controlling for student free/reduced lunch, the level-2 contextual free/reduced lunch effect of -1.68 per additional 10% of students was still significant. A random slope for the effect of free/reduced lunch also resulted in a significant improvement in model fit, $-2\Delta LL(2) = 88.3$, $p < .001$, indicating that the size of the disadvantage related to free/reduced lunch differed significantly across schools. A 95% random effects confidence interval for the student free/reduced lunch effect, calculated as fixed slope $\pm 1.96 * \text{SQRT}(\text{random slope variance})$, revealed that 95% of the schools were predicted to have lunch-related gaps between students ranging from -15.5 to -1.39 .

The extent to which school differences in the lunch-related math disadvantage could be predicted from school lunch composition was then examined by adding a cross-level intra-variable interaction between the student and school lunch predictors, as well as the quadratic effect of school lunch composition to control for a contextual interaction effect. The two new interaction effects did not significantly reduce additional total variance in math scores, change in total- $R^2 = .001$, $F(2,95) = 2.45$, $p = .091$, but were retained and interpreted given their hypothesized importance. Parameters for this final model are given in Table X [table should provide estimates, SEs, and p -values (or stars or bold font) for all model parameters, including variance parameters]. The level-1 within-school lunch effect was marginally moderated by school lunch composition (which reduced its random slope variance by 4.97%), although the moderation of the between-school and contextual effects was not significant (reducing the random intercept variance by another 3.54%). The pattern of the two interactions is depicted by the nonparallel slopes of the lines in Figure 1, indicated that the significant lunch-related disadvantage in math scores of 8.82, as found for students receiving free/reduced lunch in schools in which 30% of students received free/reduced lunch, became marginally less negative (smaller) by 0.54 for every additional 10% of students who received free/reduced lunch. In addition, the significant contextual school effect of -1.80 per 10% free/reduced lunch students (for schools with 30% free/reduced lunch students) was nonsignificantly reduced by 0.54 in free/reduced lunch students, as shown by the narrower gap between the lines on the right side relative to the left side of the x-axis. The level-2 quadratic effect, seen by the widening distance between the lines, indicated that the same contextual school effect (of -1.80 per 10% free/reduced lunch students for schools with 30% free/reduced lunch students) became nonsignificantly more negative by 1.43 for every additional 10% free/reduced lunch students (i.e., controlling for student lunch status), or that the between-school effect (of -2.68 per 10% students for schools with 30% free/reduced lunch students) became nonsignificantly more negative by 0.88 per 10% students (i.e., not controlling for student lunch status).

Figure 1: Plot of model-predicted math by free/reduced lunch status

